Math 132
Fall 2007 Final Exam

1. Calculate $\int_0^{\frac{\pi}{2}} \cos(x)\sin(x)^3\,dx$.

a) 1 b) $\frac{1}{2}$ c) $\frac{1}{3}$ d) $\frac{1}{4}$ e) $\frac{1}{5}$

f) $\frac{2}{3}$ g) $\frac{3}{4}$ h) $\frac{3}{2}$ i) $\frac{4}{3}$ j) $\frac{1}{6}$

Solution: d

```
> J := Int(cos(x)*sin(x)^3, x = 0..Pi/2);
```

$$J := \int_0^{\frac{\pi}{2}} \cos(x)\sin(x)^3\,dx$$

```
> K := student[changevar](u = sin(x), J, u);
```

$$K := \int_0^1 u^3\,du$$

```
> value(K);
```

$$\frac{1}{4}$$

2. Let $F(x) = \int_x^2 \frac{5+t^4}{\sqrt{1+t^3}}\,dt$. Calculate the derivative D(F)(2) of F at 2.

a) 4 b) 5 c) 6 d) 7 e) 8

f) –4 g) –5 h) –6 i) –7 j) –8

Solution: i

```
> F := (x) -> Int((5+t^4)/sqrt(1+t^3),t = x .. 2);
```

$$F := x \to \int_x^2 \frac{5+t^4}{\sqrt{1+t^3}}\, dt$$

```
> D(F)(x);
```

$$-\frac{5+x^4}{\sqrt{1+x^3}}$$

```
> D(F)(2);
```

$$-\frac{7\sqrt{9}}{3}$$

```
> simplify(D(F)(2));
```

$$-7$$

3. Calculate $\displaystyle\int_0^1 \frac{x}{(x+1)(x+2)}\, dx$**.**

a) $\ln\left(\frac{9}{8}\right)$ b) $\ln\left(\frac{7}{6}\right)$ c) $\ln\left(\frac{5}{4}\right)$ d) $\ln\left(\frac{4}{3}\right)$ e) $\ln\left(\frac{3}{2}\right)$

f) $\ln\left(\frac{9}{5}\right)$ g) $\ln\left(\frac{8}{3}\right)$ h) $\ln\left(\frac{9}{4}\right)$ i) $\ln\left(\frac{16}{3}\right)$ j) $\ln\left(\frac{16}{9}\right)$

Solution: a

```
> J := Int(x/(x+1)/(x+2),x = 0 .. 1);
```

$$J := \int_0^1 \frac{x}{(x+1)(x+2)}\, dx$$

```
> R := student[integrand](J);
```

$$R := \frac{x}{(x+1)(x+2)}$$

```
> PFE := convert(R, parfrac, x);
```

$$PFE := -\frac{1}{x+1} + \frac{2}{x+2}$$

```
> antiderivative := int(PFE, x);
```

$$antiderivative := -\ln(x+1) + 2\ln(x+2)$$

```
> definiteIntegral := subs(x=1,antiderivative) -
  subs(x=0,antiderivative);
```

$$definiteIntegral := -3\ln(2) + 2\ln(3) + \ln(1)$$

```
> Answer := combine(definiteIntegral, ln);
```

$$Answer := -\ln\left(\frac{8}{9}\right)$$

4. Calculate $\displaystyle\int_0^1 \frac{8x^2+2x+6}{(1+x)(1+x^2)}\,dx$.

a) $\frac{1}{4}\ln(2)$ **b)** $\frac{1}{2}\ln(2)$ **c)** $\ln(2)$ **d)** $2\ln(2)$ **e)** $3\ln(2)$

f) $4\ln(2)$ **g)** $5\ln(2)$ **h)** $6\ln(2)$ **i)** $7\ln(2)$ **j)** $8\ln(2)$

Solution: i

```
> J := Int((8*x^2+2*x+6)/(1+x)/(1+x^2),x = 0 .. 1);
```

$$J := \int_0^1 \frac{8x^2+2x+6}{(x+1)(1+x^2)}\,dx$$

```
> R := student[integrand](J);
```

$$R := \frac{8x^2+2x+6}{(x+1)(1+x^2)}$$

```
> PFE := convert(R, parfrac, x);
```

$$PFE := \frac{2x}{1+x^2} + \frac{6}{x+1}$$

```
> antiderivative := int(PFE, x);
```

$$antiderivative := \ln(1+x^2) + 6\ln(x+1)$$

```
> definiteIntegral := subs(x=1,antiderivative) -
  subs(x=0,antiderivative);
```

$$definiteIntegral := 7\ln(2) - 7\ln(1)$$

```
> Answer := combine(definiteIntegral, ln);
```

$$Answer := 7\ln(2)$$

5. Calculate $\displaystyle\int_1^{e} x^2 \ln(x)\,dx$.

a) $\frac{1}{3}e^3$ b) $\frac{1}{3}(2e^3-1)$ c) $\frac{1}{3}(e^3-2)$ d) $\frac{2}{3}(e^3-1)$ e) $\frac{1}{3}(2e^3+1)$

f) $\frac{1}{3}(e^3+2)$ g) $\frac{2}{3}(e^3+1)$ h) $\frac{1}{9}(2e^3+1)$ i) $\frac{1}{9}(e^3+2)$ j) $\frac{2}{9}(e^3+1)$

Solution: h

```
> J := Int(x^2*ln(x), x = 1 .. exp(1));
```

$$J := \int_1^{e} x^2 \ln(x)\,dx$$

```
> K := student[intparts](J, ln(x)); #Integration by Parts with
  u=ln(x)
```

$$K := \frac{1}{3}(e)^3 - \int_1^{e} \frac{x^2}{3}\,dx$$

```
> value(K);
```

$$\frac{2}{9}(e)^3 + \frac{1}{9}$$

6. What is the derivative of $x^{\left(\frac{1}{x}\right)}$ with respect to x at $x = \frac{1}{2}$?

a) $-\ln(2)$ b) $-\frac{1}{2}\ln(2)$ c) $1-\ln(2)$ d) $1-\frac{1}{2}\ln(2)$ e) $\ln(2)$

f) $\frac{1}{2}\ln(2)$ g) $1+\ln(2)$ h) $1+\frac{1}{2}\ln(2)$ i) $\frac{1}{4}\ln(2)$ j) $\frac{1}{4}$

Solution: g

```
> restart;
> eqn1 := f(x) = x^(1/x);
```

$$eqn1 := f(x) = x^{\left(\frac{1}{x}\right)}$$

```
> eqn2 := map(z-> simplify(ln(z), symbolic), eqn1);
```

$$eqn2 := \ln(f(x)) = \frac{\ln(x)}{x}$$

```
> eqn3 := map(z -> diff(z,x), eqn2);
```

$$eqn3 := \frac{\frac{d}{dx} f(x)}{f(x)} = -\frac{\ln(x)}{x^2} + \frac{1}{x^2}$$

```
> eqn4 := D(f)(x) = solve(eqn3, diff(f(x),x));
```

$$eqn4 := D(f)(x) = -\frac{f(x)(\ln(x) - 1)}{x^2}$$

```
> eqn5 := subs(x = 1/2, eqn4);
```

$$eqn5 := D(f)\left(\frac{1}{2}\right) = -4\, f\left(\frac{1}{2}\right)\left(\ln\left(\frac{1}{2}\right) - 1\right)$$

```
> eqn6 := subs(x = 1/2, eqn1);
```

$$eqn6 := f\left(\frac{1}{2}\right) = \frac{1}{4}$$

```
> subs(eqn6, eqn5);
```

$$D(f)\left(\frac{1}{2}\right) = \ln(2) + 1$$

7. If $y(0) = 0$ and $\frac{dy}{dx} = \cos(x)\sqrt{1 - y^2}$, then what is $y(x)$?

a) $\sin(\pi \cos(x))$ b) $\sin(\sin(x))$ c) $\cos\left(\frac{\pi \cos(x)}{2}\right)$ d) $\cos(\sin(x))$ -1 e) $\arcsin(x^2)$

f) $\arcsin(\arcsin(x))$ g) $\sin(\tan(x))$ h) $\tan(\sin(x))$ i) $\arcsin(\tan(x))$ j) $\arcsin(\arctan(x))$

Solution: b

By the Method of Separation of Variables:

```
> eqn1 := Int(1/sqrt(1-t^2),t = 0 .. y(x)) = Int(cos(t),t=0..x);
```

$$eqn1 := \int_0^{y(x)} \frac{1}{\sqrt{1-t^2}}\,dt = \int_0^x \cos(t)\,dt$$

```
> eqn2 := map(value, eqn1);
```

$$eqn2 := \arcsin(y(x)) = \sin(x)$$

```
> Answer := y(x) = solve(eqn2, y(x));
```

$$Answer := y(x) = \sin(\sin(x))$$

For those who are interested, here is how to get **MAPLE** to solve this differential equation without the user supplying any guidance:

```
> ode := diff(y(x),x)=cos(x)*sqrt(1-y(x)^2);
```

$$ode := \frac{d}{dx}\,y(x) = \cos(x)\sqrt{1-y(x)^2}$$

```
> initialCondition := y(0)=0;
```

$$initialCondition := y(0) = 0$$

```
> IVP := {ode, initialCondition};
```

$$IVP := \left\{ y(0) = 0, \frac{d}{dx}\,y(x) = \cos(x)\sqrt{1-y(x)^2} \right\}$$

```
> dsolve(IVP, y(x));
  #Using Maple's differential equation solver
```

$$y(x) = \sin(\sin(x))$$

8. Consider the following three statements about a series $\sum_{n=1}^{\infty} a_n$ with positive terms:

I: The series converges because $\lim_{n \to \infty} a_n = 0$.

II: The series converges because $\lim_{n \to \infty} \frac{a_{n+1}}{b_n} = 1.1$ and $\sum_{n=1}^{\infty} b_n$ converges.

III: The series converges because $\lim_{n \to \infty} \frac{a_{n+1}}{a_n} = 1$.

For each statement, determine whether the reasoning is correct or incorrect.

a) I: correct, II: correct, III: correct
b) I: correct, II: correct, III: incorrect
c) I: correct, II: incorrect, III: correct
d) I: correct, II: incorrect, III: incorrect
e) I: incorrect, II: correct, III: correct
f) I: incorrect, II: correct, III: incorrect
g) I: incorrect, II: incorrect, III: correct
h) I: incorrect, II: incorrect, III: incorrect
i) Wrong answer
j) Bonus wrong answer

Solution: f

I) Incorrect: When $a_n = \frac{1}{n}$ the terms of the series satisfy $\lim_{n \to \infty} a_n = 0$ but the series diverges.

II) Correct: The assertion follows from the Limit Comparison Test. It is true that the Limit Comparison Test is stated with the slightly different hypothesis

$$\lim_{n \to \infty} \frac{a_n}{b_n} = L.$$

To see why the Limit Comparison Test applies nonetheless, define the series $\sum_{n=0}^{\infty} c_n$ by $c_n = a_{n+1}$. The sum of the first N terms of $\sum_{n=0}^{\infty} c_n$ is $\sum_{n=0}^{N-1} c_n$, which equals $\sum_{n=1}^{N} a_n$. Since the series $\sum_{n=1}^{\infty} a_n$ and $\sum_{n=0}^{\infty} c_n$ have the same partial sums, they both converge or both diverge.

Since

$$\lim_{n \to \infty} \frac{c_n}{b_n} = \lim_{n \to \infty} \frac{a_{n+1}}{b_n} = 1.1$$

and $\sum_{n=1}^{\infty} b_n$ converges, we conclude from the Limit Comparison Test that $\sum_{n=0}^{\infty} c_n$ converges.

Therefore $\sum_{n=1}^{\infty} a_n$ also converges.

III) Incorrect: The limit condition is the inconclusive case of the Ratio Test. (When $a_n = \frac{1}{n}$ we have $\lim_{n \to \infty} \frac{a_{n+1}}{a_n} = 1$ but the series diverges.)

9. Consider the following three statements about a series $\sum_{n=1}^{\infty} a_n$ with positive terms:

I: The series converges because $a_n < \frac{1}{10 + \sqrt{n}}$.

II: The series diverges because $\frac{1}{n^2} < a_n$.

III: The series converges because $\lim_{n \to \infty} \frac{a_{n+1}}{a_n} = 0$.

For each statement, determine whether the reasoning is correct (C) or incorrect (F).

a) I: C, II: C, III: C
b) I: C, II: C, III: F
c) I: C, II: F, III: C
d) I: C, II: F, III: F
e) I: F, II: C, III: C
f) I: F, II: C, III: F
g) I: F, II: F, III: C
h) I: F, II: F, III: F
i) Wrong answer
j) Bonus wrong answer

Solution: g

I) Incorrect (F): The series $\sum_{n=1}^{\infty} \frac{1}{10+\sqrt{n}}$ is divergent (by comparison with the divergent p-series $\sum_{n=1}^{\infty} \frac{1}{\sqrt{n}}$.) No information about $\sum_{n=1}^{\infty} a_n$ can be deduced from $a_n < b_n$ when $\sum_{n=1}^{\infty} b_n$ is divergent.

II) Incorrect (F): The series $\sum_{n=1}^{\infty} \frac{1}{n^2}$ is convergent. No information about $\sum_{n=1}^{\infty} a_n$ can be deduced from

$b_n < a_n$ when $\sum_{n=1}^{\infty} b_n$ is convergent.

III) Correct (C): Since the limit, namely 0, is less than 1, this assertion follows from the Ratio Test.

10. Consider the three series

$$\text{I: } \sum_{n=0}^{\infty} \frac{n^5}{3^n}, \qquad \text{II: } \sum_{n=0}^{\infty} \frac{10^n}{\sqrt{n!}}, \qquad \text{and} \qquad \text{III: } \sum_{n=2}^{\infty} \frac{1}{n \ln(n)}$$

and the statements

(C) The series converges
(D) The series diverges

For each series, decide which of statements (C), (D) is correct.

a) I: C, II: C, III: C
b) I: C, II: C, III: D
c) I: C, II: D, III: C
d) I: C, II: D, III: D
e) I: D, II: C, III: C
f) I: D, II: C, III: D
g) I: D, II: D, III: C
h) I: D, II: D, III: D
i) Wrong answer
j) Bonus wrong answer

Solution: b

I)

```
> a := n -> n^5/3^n;
```

$$a := n \rightarrow \frac{n^5}{3^n}$$

```
> a(n+1), a(n);
```

$$\frac{(n+1)^5}{3^{(n+1)}}, \frac{n^5}{3^n}$$

```
> 'a(n+1)/a(n)' = a(n+1)/a(n);
```

$$\frac{\mathrm{a}(n+1)}{\mathrm{a}(n)} = \frac{(n+1)^5\, 3^n}{3^{(n+1)}\, n^5}$$

```
> 'a(n+1)/a(n)' = simplify(a(n+1)/a(n));
```

$$\frac{\mathrm{a}(n+1)}{\mathrm{a}(n)} = \frac{(n+1)^5}{3\, n^5}$$

```
> Limit('a(n+1)/a(n)', n=infinity) = limit(1/3*(n+1)^5/n^5,
  n=infinity);
```

$$\lim_{n \rightarrow \infty} \frac{\mathrm{a}(n+1)}{\mathrm{a}(n)} = \frac{1}{3}$$

Since this limit is less than 1, the Ratio Test gives convergence (C).

II)

```
> b := n -> 10^n/sqrt(n!);
```

$$b := n \rightarrow \frac{10^n}{\sqrt{n!}}$$

```
> b(n+1), b(n);
```

$$\frac{10^{(n+1)}}{\sqrt{(n+1)!}}, \frac{10^n}{\sqrt{n!}}$$

```
> 'b(n+1)/b(n)' = b(n+1)/b(n);
```

$$\frac{\mathrm{b}(n+1)}{\mathrm{b}(n)} = \frac{10^{(n+1)}\sqrt{n!}}{\sqrt{(n+1)!}\; 10^n}$$

```
> 'b(n+1)/b(n)' = 10*sqrt(n!/(n+1)!);
```

$$\frac{b(n+1)}{b(n)} = 10\sqrt{\frac{n!}{(n+1)!}}$$

```
> 'b(n+1)/b(n)' = simplify(10*sqrt(n!/(n+1)!)) assuming
  (n,posint);
```

$$\frac{b(n+1)}{b(n)} = \frac{10}{\sqrt{n+1}}$$

```
> Limit('b(n+1)/b(n)', n=infinity) = limit(10/(n+1)^(1/2),
  n=infinity);
```

$$\lim_{n \to \infty} \frac{b(n+1)}{b(n)} = 0$$

Since this limit is less than 1, the Ratio Test gives convergence (C).

III)

```
> f := x -> 1/x/ln(x);
```

$$f := x \to \frac{1}{x \ln(x)}$$

```
> Int(1/x/ln(x),x= 2 .. N) = int(1/x/ln(x),x= 2 .. N);
```

$$\int_2^N \frac{1}{x \ln(x)}\, dx = \ln(\ln(N)) - \ln(\ln(2))$$

```
> Int(1/(x*ln(x)),x = 2 .. infinity) = Limit(
  ln(ln(N))-ln(ln(2)), N = infinity);
```

$$\int_2^\infty \frac{1}{x \ln(x)}\, dx = \lim_{N \to \infty} \ln(\ln(N)) - \ln(\ln(2))$$

```
> Int(1/(x*ln(x)),x = 2 .. infinity) = limit(
  ln(ln(N))-ln(ln(2)), N = infinity);
```

$$\int_2^\infty \frac{1}{x \ln(x)}\, dx = \infty$$

The given series diverges (D) by the Integral Test.

11. **Consider the two series**

$$\text{I: } \sum_{n=1}^{\infty} (-1)^n \left(\frac{n}{1+n}\right)^n \quad \text{and} \quad \text{II: } \sum_{n=0}^{\infty} \frac{(-1)^n n^{\left(\frac{2}{3}\right)}}{1+n^{\left(\frac{4}{3}\right)}}$$

and the statements

(AC) The series converges absolutely
(CC) The series converges conditionally
(D) The series diverges

For each series, decide which of statements (AC), (CC), (D) is correct.

a) I: AC, II: AC
b) I: AC, II: CC
c) I: AC, II: D
d) I: CC, II: AC
e) I: CC, II: CC
f) I: CC, II: D
g) I: D, II: AC
h) I: D, II: CC
i) I: D, II: D
j) Wrong answer

Solution: h

```
> a := n -> (n/(1+n))^n;
```

$$a := n \rightarrow \left(\frac{n}{n+1}\right)^n$$

We studied the limit of this sequence in conjunction with continuous compounding.

```
> limit(a(n), n = infinity);
```

$$e^{(-1)}$$

Since this number is not 0, it follows from the divergence test that the series diverges (D).

II)

```
> a := n -> n^(2/3)/(1+n^(4/3));
```

$$a := n \to \frac{n^{(2/3)}}{1+n^{(4/3)}}$$

```
> b := n -> 1/n^(2/3); #Captures the size of a(n)
```

$$b := n \to \frac{1}{n^{(2/3)}}$$

```
> limit(a(n)/b(n), n = infinity);
```

$$1$$

Since this number is not 0 and not ∞, we conclude from the Limit Comparison Test that the series $\sum_{n=1}^{\infty} \frac{n^{\left(\frac{2}{3}\right)}}{1+n^{\left(\frac{4}{3}\right)}}$ has the same behavior as the series $\sum_{n=1}^{\infty} \frac{1}{n^{\left(\frac{2}{3}\right)}}$, which is divergent since it is a p-series with $p = \frac{2}{3} \le 1$. As a result, we conclude that the given series is not absolutely convergent. Since the given series converges by the Alternating Series Test, we conclude that it is conditionally convergent (CC).

12. Consider the two series

I: $\sum_{n=0}^{\infty} \frac{1}{n^{\pi}}$ **and II:** $\sum_{n=0}^{\infty} \frac{n!}{10^{(100\,n)}}$

and the statements

(C) The Ratio Test establishes convergence
(D) The Ratio Test establishes divergence
(F) The Ratio Test is not conclusive.

Apply the Ratio Test to series I and II and for each, decide which of statements (C), (D), (F) is correct.

a) I: C, II: C
b) I: C, II: D
c) I: C, II: F
d) I: D, II: C
e) I: D, II: D
f) I: D, II: F
g) I: F, II: C
h) I: F, II: D
i) I: F, II: F
j) Wrong answer

Solution: h

I

```
> a := n -> 1/n^Pi;
```

$$a := n \to \frac{1}{n^{\pi}}$$

```
> Limit(a(n+1)/a(n), n = infinity) = limit(a(n+1)/a(n), n = infinity);
```

$$\lim_{n \to \infty} \frac{n^{\pi}}{(n+1)^{\pi}} = 1$$

Because this is equal to 1, the Ratio Test is not conclusive (F).

II

```
> a := n -> n!/(10^(100*n));
```

$$a := n \to \frac{n!}{10^{(100\, n)}}$$

```
> r := simplify(a(n+1)/a(n));
```

$r := n$ / 100000000000000000000000000000000000000000000000000000000000000000000\
00000000000000000000000000000000 + 1 / 10000000000000000000000000000000000000\
000000000000000000000000000000000000000000000000000000000000000

```
> limit(r, n = infinity);
```

$$\infty$$

Because this limit is greater than 1, the Ratio Test establishes divergence (D).

13. Consider the two series

$$\text{I:}\quad \sum_{n=0}^{\infty} \left(\frac{1+n^3}{10+100\,n^2+n^3}\right)^n \qquad \text{and} \qquad \text{II:}\quad \sum_{n=1}^{\infty} \left(\frac{3+n}{3\,n}\right)^n$$

and the statements

(C) The Root Test establishes convergence
(D) The Root Test establishes divergence
(F) The Root Test is not conclusive.

Apply the Root Test to series I and II and for each, decide which of statements (C), (D), (F) is correct.

a) I: C, II: C
b) I: C, II: D
c) I: C, II: F
d) I: D, II: C
e) I: D, II: D
f) I: D, II: F
g) I: F, II: C
h) I: F, II: D
i) I: F, II: F
j) Wrong answer

Solution: g

I

```
> a := n -> ((1+n^3)/(10+100*n^2+n^3))^n;
```

$$a := n \to \left(\frac{1+n^3}{10+100\,n^2+n^3}\right)^n$$

```
> r := combine(a(n)^(1/n), symbolic);
```

$$r := \frac{1+n^3}{10+100\,n^2+n^3}$$

```
> Limit(a(n)^(1/n), n = infinity) = limit(r, n = infinity);
```

$$\lim_{n\to\infty}\left(\left(\frac{1+n^3}{10+100\,n^2+n^3}\right)^n\right)^{\left(\frac{1}{n}\right)} = 1$$

Because this is equal to 1, the Root Test is not conclusive (F).

II

```
> a := n -> ((3+n)/3/n)^n;
```

$$a := n \to \left(\frac{1}{3}\,\frac{3+n}{n}\right)^n$$

```
> r := combine(a(n)^(1/n), symbolic);
```

$$r := \frac{1+\frac{n}{3}}{n}$$

```
> Limit(a(n)^(1/n), n = infinity) = limit(r, n = infinity);
```

$$\lim_{n\to\infty}\left(\left(\frac{3+n}{3\,n}\right)^n\right)^{\left(\frac{1}{n}\right)} = \frac{1}{3}$$

Because this is less than 1, the Root Test establishes convergence (C).

14. Let $f(x) = \frac{1}{72}x^3\,e^{(2\,x^2)}$. What is $f^{(\,7\,)}(\,0\,)$?

a) 20 b) 40 c) 60 d) 80 e) 100
f) 120 g) 140 h) 160 i) 180 j) 200

Solution: g

```
> f := x -> x^3*exp(2*x^2)/72;
```

$$f := x \to \frac{1}{72} x^3 \mathrm{e}^{(2 x^2)}$$

```
> Maclaurin := series(f(x), x = 0, 10);
```

$$Maclaurin := \frac{1}{72} x^3 + \frac{1}{36} x^5 + \frac{1}{36} x^7 + \frac{1}{54} x^9 + O(x^{11})$$

```
> p := convert(Maclaurin, polynom);
```

$$p := \frac{x^3}{72} + \frac{x^5}{36} + \frac{x^7}{36} + \frac{x^9}{54}$$

```
> Answer = 7!*coeff(p, x^7);
```

$$Answer = 140$$

By direct calculation (not recommended!):

```
> (D@@7)(f)(x);
```

$$140\, \mathrm{e}^{(2 x^2)} + 3080 x^2 \mathrm{e}^{(2 x^2)} + 8960 x^4 \mathrm{e}^{(2 x^2)} + \frac{23296}{3} x^6 \mathrm{e}^{(2 x^2)} + \frac{7168}{3} x^8 \mathrm{e}^{(2 x^2)} + \frac{2048}{9} x^{10} \mathrm{e}^{(2 x^2)}$$

```
> (D@@7)(f)(0);
```

$$140$$

15. Calculate $L = \lim_{x \to 0} \dfrac{120 \sin(2 x^5)}{x \cos(5 x^2) - x}$ by finding the Maclaurin series of the numerator and the Maclaurin series of the denominator. These two Maclaurin series begin with the same degree p monomial. (In other words, for the Maclaurin series for both the numerator and denominator, the coefficients of x^n are 0 for $n < p$ and the coefficients of x^p are nonzero.) What is the value of the product $p\,L$?

a) - 36 b) - 48 c) - 60 d) - 72 e) - 84
f) - 96 g) -108 h) -120 i) - 132 j) - 144

Solution: f

```
> ratio := x -> 120*sin(2*x^5)/(x*cos(5*x^2)-x);
```

$$ratio := x \rightarrow \frac{120 \sin(2 x^5)}{x \cos(5 x^2) - x}$$

```
> Limit(ratio(x), x = 0) = limit(ratio(x), x = 0);
```

$$\lim_{x \to 0} \frac{120 \sin(2 x^5)}{x \cos(5 x^2) - x} = \frac{-96}{5}$$

```
> series(numer(ratio(x)), x = 0, 10);
  series(denom(ratio(x)), x = 0, 10);
```

$$240 x^5 + \mathrm{O}(x^{10})$$

$$-\frac{25}{2} x^5 + \frac{625}{24} x^9 + \mathrm{O}(x^{11})$$

```
> Answer = 5*(-96/5);
```

$$Answer = -96$$

16. Calculate the interval of convergence of $\sum_{n=0}^{\infty} \frac{(-1)^n (x+3)^n}{\sqrt{n+1}\ 4^n}$. **Let R be the radius of convergence. You will need to calculate the sum of four integers and it might help to record them as you go.**

Let c be the base point of the power series. (c = _________)

Set ρ = R if R is an integer and -1 otherwise. (ρ = _________)

Set $\sigma = 1$ if the left endpoint belongs to the interval of convergence and 0 otherwise. (σ = _________)

Set $\tau = 3$ if the right endpoint belongs to the interval of convergence and 0 otherwise. (τ = _________)

What is the value of $c + \rho + \sigma + \tau$?

a) -4 b) -3 c) -2 d) 2 e) 3
f) 4 g) 7 h) 8 i) 10 j) 11

Solution: f

```
> c := -3;
```

$$c := -3$$

```
> a := n -> (-1)^n/sqrt(n+1)/4^n;
```

$$a := n \rightarrow \frac{(-1)^n}{\sqrt{n+1}\ 4^n}$$

```
> eqn := R = Limit(abs(a(n)/a(n+1)), n = infinity);
```

$$eqn := R = \lim_{n \rightarrow \infty} \left| \frac{(-1)^n \sqrt{n+2}\ 4^{(n+1)}}{\sqrt{n+1}\ 4^n\ (-1)^{(n+1)}} \right|$$

```
> R := limit(sqrt(n+2)*4^(n+1)/sqrt(n+1)/4^n, n = infinity);
```

$$R = 4$$

```
> rho := 4;
```

$$\rho := 4$$

The left endpoint is $c - R$, or -7. Substituting $x = -7$ results in the series $\sum_{n=0}^{\infty} \frac{(-1)^n (-1)^n}{\sqrt{n+1}}$, or $\sum_{n=0}^{\infty} \frac{1}{\sqrt{n+1}}$, which is a divergent p-series $\left(p = \frac{1}{2} \leq 1\right)$.

```
> sigma := 0;
```

$$\sigma := 0$$

The right endpoint is $c + R$, or 1. Substituting $x = 1$ results in the series $\sum_{n=0}^{\infty} \frac{(-1)^n}{\sqrt{n+1}}$, which is a convergent alternating series because $\frac{1}{\sqrt{n+1}}$ decreases to 0.

```
> tau := 3;
```

$$\tau := 3$$

Finally,

```
> c+rho+sigma+tau;
```

$$4$$

17. Let $\mathrm{T}(x)$ be the degree 2 Taylor polynomial of $\ln(x)$ with base point 2. What is $\mathrm{T}(3)-\ln(2)$?

a) $\frac{1}{8}$ b) $\frac{1}{4}$ c) $\frac{3}{8}$ d) $\frac{1}{2}$ e) $\frac{5}{8}$
f) $\frac{3}{4}$ g) $\frac{7}{8}$ h) 1 i) $\frac{5}{4}$ j) $\frac{3}{2}$

Solution: c

```
> c := 2:
  f := ln:
> T := x -> sum((D@@n)(f)(c)/n!*(x-c)^n, n = 0 .. 2);
```

$$T := x \rightarrow \sum_{n=0}^{2} \frac{(\mathrm{D}^{(n)})(f)(c)\,(x-c)^n}{n!}$$

```
> T(x); #The degree 2 Taylor polynomial of ln(x) with base point 2
```

$$\ln(2)+\frac{x}{2}-1-\frac{(x-2)^2}{8}$$

```
> T(3)-ln(2);
```

$$\frac{3}{8}$$

18. To approximate $\int_0^{\frac{1}{2}} \frac{\arctan(x)-x}{x^2}\,dx$, the Maclaurin series of $\arctan(x)$ (and, from that, the Maclaurin series of the integrand) is used. An alternating series for the (exact) value S of the definite integral results. An approximation to S is obtained by using the minimum

number of terms that, by the Alternating Series Test, guarantee an absolute error less than 0.001. What is the approximation?

a) $-\frac{96}{2401}$ b) $-\frac{209}{5376}$ c) $-\frac{19}{480}$ d) $-\frac{13}{336}$ e) $-\frac{2089}{53760}$

f) $-\frac{25069}{645120}$ g) $-\frac{131}{3360}$ h) $-\frac{523}{13440}$ i) $-\frac{3}{80}$ j) $-\frac{37}{960}$

Solution: j

```
> series(arctan(t), t=0, 11); #The starting point.
                              Subtract t then divide by t^2.
```

$$t-\frac{1}{3}t^3+\frac{1}{5}t^5-\frac{1}{7}t^7+\frac{1}{9}t^9+O(t^{11})$$

```
> Maclaurin := series((arctan(t)-t)/t^2, t=0, 15); #More terms
  than needed
```

$$Maclaurin := -\frac{1}{3}t+\frac{1}{5}t^3-\frac{1}{7}t^5+\frac{1}{9}t^7-\frac{1}{11}t^9+\frac{1}{13}t^{11}+O(t^{13})$$

```
> p := convert(Maclaurin, polynom);
```

$$p := -\frac{t}{3}+\frac{t^3}{5}-\frac{t^5}{7}+\frac{t^7}{9}-\frac{t^9}{11}+\frac{t^{11}}{13}$$

```
> eqn := Int((arctan(t)-t)/(t^2),t = 0 .. x) = int(p, t=0 .. x) +
  `...`;
```

$$eqn := \int_0^x \frac{\arctan(t)-t}{t^2}\,dt = -\frac{x^2}{6}+\frac{x^4}{20}-\frac{x^6}{42}+\frac{x^8}{72}-\frac{x^{10}}{110}+\frac{x^{12}}{156}+\ldots$$

We need to identify the first summand on the right side of this equation that evaluates to less than 0.001 when $x=\frac{1}{2}$.

The term $\frac{x^2}{6}$ clearly does not do the job. Next,

```
> subs(x=1/2, x^4/20);
```

$$\frac{1}{320}$$

is also not small enough. Next,

```
> subs(x=1/2, x^6/42);
```

$$\frac{1}{2688}$$

shows that this is the first term that is less than the acceptable error. We therefore add the terms up to but not including this one.

```
> Answer := subs(x=1/2, -1/6*x^2+1/20*x^4);
```

$$Answer := \frac{-37}{960}$$

As a verification, **MAPLE**'s (exact) integration is:

```
> definiteIntegral := int((arctan(x)-x)/(x^2),x = 0 .. 1/2);
```

$$definiteIntegral := -2\arctan\left(\frac{1}{2}\right) - \frac{1}{2}\ln(5) + \ln(2) + 1$$

The absolute error is:

```
> abs(evalf(Answer-definiteIntegral));
```

$$0.0003253264$$

19. What is the coefficient of x^5 in the Maclaurin series of $\frac{8x}{4-x^2}$?

a) $\frac{1}{16}$ **b)** $\frac{-1}{16}$ **c)** $\frac{1}{8}$ **d)** $-\frac{1}{8}$ **e)** $\frac{1}{4}$

f) $-\frac{1}{4}$ **g)** $\frac{1}{2}$ **h)** $-\frac{1}{2}$ **i)** 2 **j)** -2

Solution: c

Method 1: The geometric series way (as intended):

We write $\dfrac{8x}{4-x^2} = 2x\dfrac{1}{1-u}$ where $u = \dfrac{x^2}{4}$.

Thus

$$\frac{8x}{4-x^2} = 2x(1+u+u^2+u^3+\ldots)$$

$$\frac{8x}{4-x^2} = 2x\left(1+\frac{x^2}{4}+\frac{x^4}{16}+\ldots\right) = 2x+\frac{x^3}{2}+\frac{x^5}{8}+\ldots.$$

The coefficient of x^5 is $\dfrac{1}{8}$.

Method 2: The brute force computational way (not recommended!):

```
> f := x -> 8*x/(4-x^2);
```

$$f := x \rightarrow \frac{8x}{4-x^2}$$

```
> (D@@5)(f)(x);
```

$$\frac{46080x^4}{(4-x^2)^5}+\frac{17280x^2}{(4-x^2)^4}+\frac{960}{(4-x^2)^3}+\frac{30720x^6}{(4-x^2)^6}$$

```
> Answer = (D@@5)(f)(0)/5!;
```

$$Answer = \frac{1}{8}$$

The **Maple** way:

```
> Maclaurin := series(8*x/(4-x^2),x=0,8);
  p := convert(Maclaurin, polynom);
  Answer := coeff(p, x^5);
```

$$Maclaurin := 2\,x + \frac{1}{2}x^3 + \frac{1}{8}x^5 + \frac{1}{32}x^7 + O(x^9)$$

$$p := 2\,x + \frac{x^3}{2} + \frac{x^5}{8} + \frac{x^7}{32}$$

$$Answer := \frac{1}{8}$$

20. What is the coefficient of x^4 in the Maclaurin series of $\dfrac{1}{(1+x^2)^{\left(\frac{1}{3}\right)}}$?

a) $-\frac{1}{9}$ **b)** $\frac{1}{9}$ **c)** $-\frac{1}{6}$ **d)** $\frac{1}{6}$ **e)** $-\frac{2}{9}$

f) $\frac{2}{9}$ **g)** $-\frac{1}{3}$ **h)** $\frac{1}{3}$ **i)** $-\frac{2}{3}$ **j)** $\frac{2}{3}$

Solution: f

Method 1: The Newton way (as intended):

In **MAPLE**, the number "α choose n", namely

$$\frac{\alpha\,(\alpha-1)\ldots(\alpha-n+1)}{n!},$$

is written as **binomial(alpha, n).**

```
> NewtonFormula := (1+u)^alpha = Sum(binomial(alpha,n)*u^n, n = 0
  .. infinity);
```

$$NewtonFormula := (1+u)^{\alpha} = \sum_{n=0}^{\infty} \mathrm{binomial}(\alpha, n)\, u^n$$

```
> subs({u=x^2, alpha = -1/3}, NewtonFormula);
```

$$\frac{1}{(1+x^2)^{(1/3)}} = \sum_{n=0}^{\infty} \mathrm{binomial}\left(\frac{-1}{3}, n\right)(x^2)^n$$

The coefficient of x^4 (or $(x^2)^n$ for $n = 2$) is:

```
> binomial(-1/3,2);
```

$$\frac{2}{9}$$

Method 2: The brute force computational way (not recommended!):

```
> f := x -> (1+x^2)^(-1/3);
```

$$f := x \to \frac{1}{(1+x^2)^{(1/3)}}$$

```
> (D@@4)(f)(x);
```

$$\frac{4480\, x^4}{81\,(1+x^2)^{(13/3)}} - \frac{448\, x^2}{9\,(1+x^2)^{(10/3)}} + \frac{16}{3\,(1+x^2)^{(7/3)}}$$

```
> (D@@4)(f)(0);
```

$$\frac{16}{3}$$

```
> (D@@4)(f)(0)/4!;
```

$$\frac{2}{9}$$

The **Maple** way:

```
> Maclaurin := series((1+x^2)^(-1/3), x = 0, 5);
  p := convert(Maclaurin, polynom);
  Answer := coeff(p, x^4);
```

$$Maclaurin := 1 - \frac{1}{3}x^2 + \frac{2}{9}x^4 + \mathrm{O}(x^6)$$

$$p := 1 - \frac{x^2}{3} + \frac{2\,x^4}{9}$$

$$Answer := \frac{2}{9}$$